

# SWI-Prolog C-library

Jan Wielemaker  
HCS,  
University of Amsterdam  
The Netherlands  
E-mail: `wielemak@science.uva.nl`

May 21, 2008

## Abstract

This document describes commonly used foreign language extensions to SWI-Prolog distributed as a package known under the name *clib*. The package defines a number of Prolog libraries with accompanying foreign libraries.

On Windows systems, the `unix` library can only be used if the whole SWI-Prolog suite is compiled using Cywin. The `crypt` library does not support DES encryption. The other libraries have been ported to native Windows.

## Contents

<b>1</b>	<b>Introduction</b>	<b>3</b>
<b>2</b>	<b>Unix Process manipulation library</b>	<b>3</b>
<b>3</b>	<b>process.pl – Create processes and redirect I/O</b>	<b>5</b>
<b>4</b>	<b>File manipulation library</b>	<b>7</b>
<b>5</b>	<b>Socket library</b>	<b>8</b>
5.1	Server applications . . . . .	10
5.2	Client applications . . . . .	11
5.3	The stream_pool library . . . . .	11
5.4	UDP protocol support . . . . .	12
<b>6</b>	<b>CGI Support library</b>	<b>13</b>
6.1	Some considerations . . . . .	14
<b>7</b>	<b>MIME decoding library</b>	<b>14</b>
<b>8</b>	<b>Password encryption library</b>	<b>15</b>
<b>9</b>	<b>SHA1 and SHA2 Secure Hash Algorithms</b>	<b>16</b>
9.1	License terms . . . . .	17
<b>10</b>	<b>Memory files</b>	<b>17</b>
<b>11</b>	<b>Time and alarm library</b>	<b>18</b>
<b>12</b>	<b>Limiting process resources</b>	<b>19</b>
<b>13</b>	<b>Installation</b>	<b>20</b>
13.1	Unix systems . . . . .	20

## 1 Introduction

Many useful facilities offered by one or more of the operating systems supported by SWI-Prolog are not supported by the SWI-Prolog kernel distribution. Including these would enlarge the *footprint* and complicate portability matters while supporting only a limited part of the user-community.

This document describes `unix` to deal with the Unix process API, `socket` to deal with inet-domain TCP and UDP sockets, `cgi` to deal with getting CGI form-data if SWI-Prolog is used as a CGI scripting language, `crypt` to provide password encryption and verification, `sha` providing cryptographic hash functions and `memfile` providing in-memory pseudo files.

## 2 Unix Process manipulation library

The `unix` library provides the commonly used Unix primitives to deal with process management. These primitives are useful for many tasks, including server management, parallel computation, exploiting and controlling other processes, etc.

The predicates are modelled closely after their native Unix counterparts. Higher-level primitives, especially to make this library portable to non-Unix systems are desirable. Using these primitives and considering that process manipulation is not a very time-critical operation we anticipate these libraries to be developed in Prolog.

### **fork(-Pid)**

Clone the current process into two branches. In the child, *Pid* is unified to `child`. In the original process, *Pid* is unified to the process identifier of the created child. Both parent and child are fully functional Prolog processes running the same program. The processes share open I/O streams that refer to Unix native streams, such as files, sockets and pipes. Data is not shared, though on most Unix systems data is initially shared and duplicated only if one of the programs attempts to modify the data.

Unix `fork()` is the only way to create new processes and `fork/2` is a simple direct interface to it.

### **exec(+Command(...Args...))**

Replace the running program by starting *Command* using the given commandline arguments. Each command-line argument must be atomic and is converted to a string before passed to the Unix call `execvp()`.

Unix `exec()` is the only way to start an executable file executing. It is commonly used together with `fork/1`. For example to start `netscape` on an URL in the background, do:

```
run_netscape(URL) :-
    (   fork(child),
        exec(netscape(URL))
    ;   true
    ).
```

Using this code, `netscape` remains part of the process-group of the invoking Prolog process and Prolog does not wait for `netscape` to terminate. The predicate `wait/2` allows waiting for a child, while `detach_IO/0` disconnects the child as a daemon process.

**wait(-Pid, -Status)**

Wait for a child to change status. Then report the child that changed status as well as the reason. *Status* is unified with `exited(ExitCode)` if the child with pid *Pid* was terminated by calling `exit()` (Prolog `halt/[0,1]`). *ExitCode* is the return=status. *Status* is unified with `signaled(Signal)` if the child died due to a software interrupt (see `kill/2`). *Signal* contains the signal number. Finally, if the process suspended execution due to a signal, *Status* is unified with `stopped(Signal)`.

**kill(+Pid, +Signal)**

Deliver a software interrupt to the process with identifier *Pid* using software-interrupt number *Signal*. See also `on_signal/2`. The meaning of the signal numbers can be found in the Unix manual.<sup>1</sup>

**pipe(-InStream, -OutStream)**

Create a communication-pipe. This is normally used to make a child communicate to its parent. After `pipe/2`, the process is cloned and, depending on the desired direction, both processes close the end of the pipe they do not use. Then they use the remaining stream to communicate. Here is a simple example:

```
:- use_module(library(unix)).

fork_demo(Result) :-
    pipe(Read, Write),
    fork(Pid),
    (   Pid == child
->   close(Read),
        format(Write, '~q.~n',
                [hello(world)]),
        flush_output(Write),
        halt
    ;   close(Write),
        read(Read, Result),
        close(Read)
    ).
```

**dup(+FromStream, +ToStream)**

Interface to Unix `dup2()`, copying the underlying filedescriptor and thus making both streams point to the same underlying object. This is normally used together with `fork/1` and `pipe/2` to talk to an external program that is designed to communicate using standard I/O.

Both *FromStream* and *ToStream* either refer to a Prolog stream or an integer descriptor number to refer directly to OS descriptors. See also `demo/pipe.pl` in the source-distribution of this package.

**detach.IO**

This predicate is intended to create Unix daemon-processes. It preforms two actions. First

---

<sup>1</sup>kill/2 should support interrupt-names as well

of all, the I/O streams `user_input`, `user_output` and `user_error` are closed and rebound to a Prolog stream that returns end-of-file on any attempt to read and starts writing to a file named `/tmp/pl-out.pid` (where  $\langle pid \rangle$  is the process-id of the calling Prolog) on any attempt to write. This file is opened only if there is data available. This is intended for debugging purposes.<sup>2</sup> Finally, the process is detached from the current process-group and its controlling terminal.

### 3 process.pl – Create processes and redirect I/O

**Compatibility** SICStus 4

**To be done** Implement detached option in `process_create/3`

The module `library(process)` implements interaction with child processes and unifies older interfaces such as `shell/[1,2]`, `open(pipe(command), ...)` etc. This library is modelled after SICStus 4.

The main interface is formed by `process_create/3`. If the process id is requested the process must be waited for using `process_wait/2`. Otherwise the process resources are reclaimed automatically.

In addition to the predicates, this module defines a file search path (see `user:file_search_path/2` and `absolute_file_name/3`) named `path` that locates files on the system's search path for executables. E.g. the following finds the executable for `ls`:

```
?- absolute_file_name(path(ls), Path, [access(execute)]).
```

#### Incompatibilities and current limitations

- Where SICStus distinguishes between an internal process id and the OS process id, this implementation does not make this distinction. This implies that `is_process/1` is incomplete and unreliable.
- SICStus only supports ISO 8859-1 (latin-1). This implementation supports arbitrary OS multi-byte interaction using the default locale.
- The `detached(Bool)` option is implemented, but not processed. It is unclear what it is supposed to do. Disable signals in the child? Use `setsid()` to detach from the session?

**process\_create(+Exe, +Args:list, +Options)**

*[det]*

Create a new process running the file *Exe* and using arguments from the given list. *Exe* is a file specification as handed to `absolute_file_name/3`. Typically one use the `path` file alias to specify an executable file on the current `PATH`. *Args* is a list of arguments that are handed to the new process. On Unix systems, each element in the list becomes a separate argument in the new process. In Windows, the arguments are simply concatenated to form the commandline. Each argument itself is either a primitive or a list of primitives. A primitive is either atomic or a term `file(Spec)`. Using `file(Spec)`, the system inserts a filename using the OS filename conventions which is properly quoted if needed.

*Options:*

---

<sup>2</sup>More subtle handling of I/O, especially for debugging is required: communicate with the syslog daemon and optionally start a debugging dialog on a newly created (X-)terminal should be considered.

- `stdin(Spec)`
- `stdout(Spec)`
- `stderr(Spec)` Bind the standard streams of the new process. `Spec` is one of the terms below. If `pipe(Pipe)` is used, the Prolog stream is a stream in text-mode using the encoding of the default locale. The encoding can be changed using `set_stream/2`.

#### **std**

Just share with the Prolog I/O streams

#### **null**

Bind to a *null* stream. Reading from such a stream returns end-of-file, writing produces no output

#### **pipe(-Stream)**

Attach input and/or output to a Prolog stream.

- `cwd(+Directory)` Run the new process in `Directory`. `Directory` can be a compound specification, which is converted using `absolute_file_name/3`.
- `process(-PID)` Unify `PID` with the process id of the created process.
- `detached(+Bool)` If `true`, detach the process from the terminal (Unix only)
- `window(+Bool)` If `true`, create a window for the process (Windows only)

If the user specifies the `process(-PID)` option, he **must** call `process_wait/2` to reclaim the process. Without this option, the system will wait for completion of the process after the last pipe stream is closed.

If the process is not waited for, it must succeed with status 0. If not, an `process_error` is raised.

#### **Windows notes**

On Windows this call is an interface to the `CreateProcess()` API. The commandline consists of the basename of *Exe* and the arguments formed from *Args*. Arguments are separated by a single space. If all characters satisfy `iswalnum()` it is unquoted. If the argument contains a double-quote it is quoted using single quotes. If both single and double quotes appear a `domain_error` is raised, otherwise double-quotes are used.

The `CreateProcess()` API has many options. Currently only the `CREATE_NO_WINDOW` option is supported through the `window(+Bool)` option. If omitted, the default is to use this option if the application has no console. Future versions are likely to support more window specific options and replace `win_exec/2`.

#### **Examples**

First, a very simple example that behaves the same as `shell('ls -l')`, except for error handling:

```
?- process_create(path(ls), ['-l'], []).
```

**Errors** `process_error(Exe, Status)` where `Status` is one of `exit(Code)` or `killed(Signal)`. Raised if the process does not exit with status 0.

**To be done** The detach option is a no-op.

#### **process\_id(-PID)**

True if *PID* is the process id of the running Prolog process.

[det]

**deprecated** Use `current_prolog_flag(pid, PID)`

**process\_id(+Process, -PID)** [det]  
*PID* is the process id of *Process*. Given that they are united in SWI-Prolog, this is a simple unify.

**is\_process(+PID)** [semidet]  
True if *PID* might be a process. Succeeds for any positive integer.

**process\_release(+PID)**  
Release process handle. In this implementation this is the same as `process_wait(PID, _)`.

**process\_wait(+PID, -Status)** [det]

**process\_wait(+PID, -Status, +Options)** [det]  
True if *PID* completed with *Status*. This call normally blocks until the process is finished.  
*Options*:

**timeout(+Timeout)**  
Default: `infinite`. If this option is a number, the waits for a maximum of *Timeout* seconds and unifies *Status* with `timeout` if the process does not terminate within *Timeout*. In this case *PID* is *not* invalidated. On Unix systems only `timeout 0` and `infinite` are supported. A 0-value can be used to poll the status of the process.

**release(+Bool)**  
Do/do not release the process. We do not support this flag and a `domain_error` is raised if `release(false)` is provided.

**process\_kill(+PID)** [det]

**process\_kill(+PID, +Signal)** [det]  
Send signal to process *PID*. Default is `term`. *Signal* is an integer, Unix signal name (e.g. `SIGSTOP`) or the more Prolog friendly variation one gets after removing `SIG` and downcase the result: `stop`. On Windows systems, *Signal* is ignored and the process is terminated using the `TerminateProcess()` API. On Windows systems *PID* must be obtained from `process_create/3`, while any *PID* is allowed on Unix systems.

**Compatibility** SICStus does not accept the prolog friendly version. We choose to do so for compatibility with `on_signal/3`.

## 4 File manipulation library

The `files` library provides additional operations on files from SWI-Prolog. It is currently very incomplete.

**set\_time\_file(+File, -OldTimes, +NewTimes)**

Query and set POSIX time attributes of a file. Both *OldTimes* and *NewTimes* are lists of option-terms. Times are represented in SWI-Prolog's standard floating point numbers. New times may be specified as `now` to indicate the current time. Defined options are:

**access(Time)**  
Describes the time of last access of the file. This value can be read and written.

**modified(*Time*)**

Describes the time the contents of the file was last modified. This value can be read and written.

**changed(*Time*)**

Describes the time the file-structure itself was changed by adding (`link()`) or removing (`unlink()`) names.

Here are some example queries. The first retrieves the access-time, while the second sets the last-modified time to the current time.

```
?- set_time_file(foo, [access(Access)], []).
?- set_time_file(foo, [], [modified(now)]).
```

## 5 Socket library

The `socket` library provides TCP and UDP inet-domain sockets from SWI-Prolog, both client and server-side communication. The interface of this library is very close to the Unix socket interface, also supported by the MS-Windows *winsock* API. SWI-Prolog applications that wish to communicate with multiple sources have three options:

1. Use I/O multiplexing based on `wait_for_input/3`. On Windows systems this can only be used for sockets, not for general (device-) file handles.
2. Use multiple threads, handling either a single blocking socket or a pool using I/O multiplexing as above.
3. Using XPCE's class *socket* which synchronises socket events in the GUI event-loop.

**tcp\_socket(-*SocketId*)**

Creates an INET-domain stream-socket and unifies an identifier to it with *SocketId*. On MS-Windows, if the socket library is not yet initialised, this will also initialise the library.

**tcp\_close\_socket(+*SocketId*)**

Closes the indicated socket, making *SocketId* invalid. Normally, sockets are closed by closing both stream handles returned by `open_socket/3`. There are two cases where `tcp_close_socket/1` is used because there are no stream-handles:

- After `tcp_accept/3`, the server does a `fork/1` to handle the client in a sub-process. In this case the accepted socket is not longer needed from the main server and must be discarded using `tcp_close_socket/1`.
- If, after discovering the connecting client with `tcp_accept/3`, the server does not want to accept the connection, it should discard the accepted socket immediately using `tcp_close_socket/1`.

**tcp\_open\_socket(+*SocketId*, -*InStream*, -*OutStream*)**

Open two SWI-Prolog I/O-streams, one to deal with input from the socket and one with output to the socket. If `tcp_bind/2` has been called on the socket. *OutStream* is useless and will not be created. After closing both *InStream* and *OutStream*, the socket itself is discarded.



**tcp\_bind(+Socket, ?Port)**

Bind the socket to *Port* on the current machine. This operation, together with `tcp_listen/2` and `tcp_accept/3` implement the *server*-side of the socket interface. If *Port* is unbound, the system picks an arbitrary free port and unifies *Port* with the selected port number. *Port* is either an integer or the name of a registered service.

**tcp\_listen(+Socket, +Backlog)**

Tells, after `tcp_bind/2`, the socket to listen for incoming requests for connections. *Backlog* indicates how many pending connection requests are allowed. Pending requests are requests that are not yet acknowledged using `tcp_accept/3`. If the indicated number is exceeded, the requesting client will be signalled that the service is currently not available. A suggested default value is 5.

**tcp\_accept(+Socket, -Slave, -Peer)**

This predicate waits on a server socket for a connection request by a client. On success, it creates a new socket for the client and binds the identifier to *Slave*. *Peer* is bound to the IP-address of the client.

**tcp\_connect(+Socket, +Host:+Port)**

Client-interface to connect a socket to a given *Port* on a given *Host*. *Port* is either an integer or the name of a registered service. After successful completion, `tcp_open_socket/3` can be used to create I/O-Streams to the remote socket. The fragment below connects to the `http://www.swi-prolog.org` using the service name instead of the hardcoded number '80'.

```
tcp_socket(Socket),
tcp_connect(Socket, 'www.swi-prolog.org':http),
```

**tcp\_setopt(+Socket, +Option)**

Set options on the socket. Defined options are:

**reuseaddr**

Allow servers to reuse a port without the system being completely sure the port is no longer in use.

**broadcast**

UDP sockets only: broadcast the package to all addresses matching the address. The address is normally the address of the local subnet (i.e. 192.168.1.255). See `udp_send/4`.

**dispatch(Bool)**

In GUI environments (using XPCE or the Windows `plwin.exe` executable) this flag defines whether or not any events are dispatched on behalf of the user interface. Default is `true`. Only very specific situations require setting this to `false`.

**tcp\_fcntl(+Stream, +Action, ?Argument)**

Interface to the Unix `fcntl()` call. Currently only suitable to deal switch stream to non-blocking mode using:

```
...
tcp_fcntl(Stream, setfl. nonblock),
...
```

As of SWI-Prolog 3.2.4, handling of non-blocking stream is supported. An attempt to read from a non-blocking stream returns -1 (or `end_of_file` for `read/1`), but `at_end_of_stream/1` fails. On actual end-of-input, `at_end_of_stream/1` succeeds.

#### **tcp\_host\_to\_address(?HostName, ?Address)**

Translate between a machines host-name and it's (IP-)address. If *HostName* is an atom, it is resolved using `gethostbyname()` and the IP-number is unified to *Address* using a term of the format `ip(Byte1, Byte2, Byte3, Byte4)`. Otherwise, if *Address* is bound to a `ip/4` term, it is resolved by `gethostbyaddr()` and the canonical hostname is unified with *HostName*.

#### **gethostname(-Hostname)**

Return the official fully qualified name of this host. This is achieved by calling `gethostname()` followed by `gethostbyname()` and return the official name of the host (`h_name`) of the structure returned by the latter function.

## **5.1 Server applications**

The typical sequence for generating a server application is defined below:

```
create_server(Port) :-
    tcp_socket(Socket),
    tcp_bind(Socket, Port),
    tcp_listen(Socket, 5),
    tcp_open_socket(Socket, AcceptFd, _),
    <dispatch>
```

There are various options for *<dispatch>*. One is to keep track of active clients and server-sockets using `wait_for_input/3`. If input arrives at a server socket, use `tcp_accept/3` and add the new connection to the active clients. Otherwise deal with the input from the client. Another is to use (Unix) `fork/1` to deal with the client in a separate process.

Using `fork/1`, *<dispatch>* may be implemented as:

```
dispatch(AcceptFd) :-
    tcp_accept(AcceptFd, Socket, _Peer),
    fork(Pid)
    (   Pid == child
    ->  tcp_open_socket(Socket, In, Out),
        handle_service(In, Out),
        close(In),
        close(Out),
        halt
    ;   tcp_close_socket(Socket)
    ),
    dispatch(AcceptFd).
```

## 5.2 Client applications

The skeleton for client-communication is given below.

```
create_client(Host, Port) :-
    tcp_socket(Socket),
    tcp_connect(Socket, Host:Port),
    tcp_open_socket(Socket, ReadFd, WriteFd),
    <handle I/O using the two streams>
    close(ReadFd),
    close(WriteFd).
```

To deal with timeouts and multiple connections, `wait_for_input/3` and/or non-blocking streams (see `tcp_fcntl/3`) can be used.

## 5.3 The stream\_pool library

The `stream_pool` library dispatches input from multiple streams based on `wait_for_input/3`. It is part of the `clib` package as it is used most of the time together with the `socket` library. On non-Unix systems it often can only be used with socket streams.

With SWI-Prolog 5.1.x, multi-threading often provides a good alternative to using this library. In this schema one thread watches the listening socket waiting for connections and either creates a thread per connection or processes the accepted connections with a pool of *worker threads*. The library `http/thread.httpd` provides an example realising a mult-threaded HTTP server.

### **add\_stream\_to\_pool(+Stream, :Goal)**

Add *Stream*, which must be an input stream and —on non-unix systems— connected to a socket to the pool. If input is available on *Stream*, *Goal* is called.

### **delete\_stream\_from\_pool(+Stream)**

Delete the given stream from the pool. Succeeds, even if *Stream* is no member of the pool. If *Stream* is unbound the entire pool is emptied but unlike `close_stream_pool/0` the streams are not closed.

### **close\_stream\_pool**

Empty the pool, closing all streams that are part of it.

### **dispatch\_stream\_pool(+TimeOut)**

Wait for maximum of *TimeOut* for input on any of the streams in the pool. If there is input, call the *Goal* associated with `add_stream_to_pool/2`. If *Goal* fails or raises an exception a message is printed. *TimeOut* is described with `wait_for_input/3`.

If *Goal* is called, there is *some* input on the associated stream. *Goal* must be careful not to block as this will block the entire pool.<sup>3</sup>

### **stream\_pool\_main\_loop**

Calls `dispatch_stream_pool/1` in a loop until the pool is empty.

Below is a very simple example that reads the first line of input and echos it back.

---

<sup>3</sup>This is hard to achieve at the moment as none of the Prolog read-commands provide for a timeout.

```

:- use_module(library(streampool)).

server(Port) :-
    tcp_socket(Socket),
    tcp_bind(Socket, Port),
    tcp_listen(Socket, 5),
    tcp_open_socket(Socket, In, _Out),
    add_stream_to_pool(In, accept(Socket)),
    stream_pool_main_loop.

accept(Socket) :-
    tcp_accept(Socket, Slave, Peer),
    tcp_open_socket(Slave, In, Out),
    add_stream_to_pool(In, client(In, Out, Peer)).

client(In, Out, _Peer) :-
    read_line_to_codes(In, Command),
    close(In),
    format(Out, 'Please to meet you: ~s~n', [Command]),
    close(Out),
    delete_stream_from_pool(In).

```

## 5.4 UDP protocol support

The current library provides limited support for UDP packets. The UDP protocol is a *connection-less* and *unreliable* datagram based protocol. That means that messages sent may or may not arrive at the client side and may arrive in a different order as they are sent. UDP messages are often used for streaming media or for service discovery using the broadcasting mechanism.

### **udp\_socket(-Socket)**

Similar to `tcp_socket/1`, but create a socket using the `SOCK_DGRAM` protocol, ready for UDP connections.

### **udp\_receive(+Socket, -Data, -From, +Options)**

Wait for and return the next datagram. The data is returned as a Prolog string object (see `string_to_list/2`). *From* is a term of the format `ip(A,B,C,D):Port` indicating the sender of the message. *Socket* can be waited for using `wait_for_input/3`. Defined *Options*:

#### **as(+Type)**

Defines the returned term-type. *Type* is one of `atom`, `codes` or `string` (default).

The typical sequence to receive UDP data is:

```

receive(Port) :-
    udp_socket(S),
    tcp_bind(S, Port),
    repeat,

```

```

udp_receive(Socket, Data, From, [as(atom)]),
format('Got ~q from ~q~n', [Data, From]),
fail.

```

### **udp\_send(+Socket, +Data, +To, +Options)**

Send a UDP message. *Data* is a string, atom or code-list providing the data. *To* is an address of the form *Host:Port* where *Host* is either the hostname or a term *ip/4*. *Options* is currently unused.

A simple example to send UDP data is:

```

send(Host, Port, Message) :-
    udp_socket(S),
    udp_send(S, Message, Host:Port, []),
    tcp_close_socket(S).

```

A broadcast is achieved by using `tcp_setopt(Socket, broadcast)` prior to sending the datagram and using the local network broadcast address as a *ip/4* term.

The normal mechanism to discover a service on the local network is for the client to send a broadcast message to an agreed port. The server receives this message and replies to the client with a message indicating further details to establish the communication.

## **6 CGI Support library**

This is currently a very simple library, providing support for obtaining the form-data for a CGI script:

### **cgi\_get\_form(-Form)**

Decodes standard input and the environment variables to obtain a list of arguments passed to the CGI script. This predicate both deals with the CGI **GET** method as well as the **POST** method. If the data cannot be obtained, an `existence_error` exception is raised.

Below is a very simple CGI script that prints the passed parameters. To test it, compile this program using the command below, copy it to your `cgi-bin` directory (or make it otherwise known as a CGI-script) and make the query `http://myhost.mydomain/cgi-bin/cgidemo?hello=world`

```
% pl -o cgidemo --goal=main --toplevel=halt -c cgidemo.pl
```

```
:- use_module(library(cgi)).
```

```

main :-
    set_stream(current_output, encoding(utf8)),
    cgi_get_form(Arguments),
    format('Content-type: text/html; charset=UTF-8~n~n', []),
    format('<HTML>~n', []).

```

```

format('<HEAD>~n', []),
format('<TITLE>Simple SWI-Prolog CGI script</TITLE>~n', []),
format('</HEAD>~n~n', []),
format('<BODY>~n', []),
format('<P>', []),
print_args(Arguments),
format('</BODY>~n</HTML>~n', []).

print_args([]).
print_args([A0|T]) :-
    A0 =.. [Name, Value],
    format('<B>~w</B>=<EM>~w</EM><BR>~n', [Name, Value]),
    print_args(T).

```

## 6.1 Some considerations

Printing an HTML document using `format/2` is not really a neat way of producing HTML. A high-level alternative is provided by `http/html_write` from the HTTP library.

The CGI standard is very unclear about handling Unicode data. The above two declarations ensure the CGI script will send all data in UTF-8 and thus provide full support of Unicode. It is assumed that browsers generally send form-data using the same encoding as the page in which the form appears, UTF-8 or ISO Latin-1. The current version of `cgi_get_form/2` assumes the CGI data is in UTF-8.

## 7 MIME decoding library

MIME (Multipurpose Internet Mail Extensions) is a format for serializing multiple typed data objects. It was designed for E-mail, but it is also used for other applications such as packaging multiple values using the HTTP POST request on web-servers. Double Precision, Inc. has produced the C-libraries `rfc822` (mail) and `rfc2045` (MIME) for decoding and manipulating MIME messages. The `mime` library is a Prolog wrapper around the `rfc2045` library for decoding MIME messages.

The general name 'mime' is used for this library as it is anticipated to add MIME-creation functionality to this library.

Currently the `mime` library defines one predicate:

**mime\_parse**(*Data*, *Parsed*)

Parse *Data* and unify the result to *Parsed*. *Data* is one of:

**stream**(*Stream*)

Parse the data from *Stream* upto the end-of-file.

**stream**(*Stream*, *Length*)

Parse a maximum of *Length* characters from *Stream* or upto the end-of-file, whichever comes first.

*Text*

Atoms, strings, code- and character lists are treated as valid sources of data.

*Parsed* is a tree structure of `mime(Attributes, Data, PartList)` terms. Currently either *Data* is the empty atom or *PartList* is an empty list.<sup>4</sup> *Data* is an atom holding the message data. The library automatically decodes *base64* and *quoted-printable* messages. See also the `transfer_encoding` attribute below.

*PartList* is a list of `mime/3` terms. *Attributes* is a list holding a subset of the following arguments. For details please consult the RFC2045 document.

**type(*Atom*)**

Denotes the Content-Type, how the *Data* should be interpreted.

**transfer\_encoding(*Atom*)**

How the *Data* was encoded. This is not very interesting as the library decodes the content of the message.

**character\_set(*Atom*)**

The character set used for text data. Note that SWI-Prolog's capabilities for character-set handling are limited.

**language(*Atom*)**

Language in which the text-data is written.

**id(*Atom*)**

Identifier of the message-part.

**description(*Atom*)**

Descriptive text for the *Data*.

**disposition(*Atom*)**

Where the data comes from. The current library only deals with 'inline' data.

**name(*Atom*)**

Name of the part.

**filename(*Atom*)**

Name of the file the data should be stored in.

## 8 Password encryption library

The `crypt` library defines `crypt/2` for encrypting and testing passwords. The `clib` package also provides cryptographic hashes as described in section 9

**crypt(+*Plain*, ?*Encrypted*)**

This predicate can be used in three modes. To test whether a password matches an encrypted version thereof, simply run with both arguments fully instantiated. To generate a default encrypted version of *Plain*, run with unbound *Encrypted* and this argument is unified to a list of character codes holding an encrypted version.

The library supports two encryption formats: traditional Unix DES hashes (Unix only) and FreeBSD compatible MD5 hashes (all platforms). MD5 hashes start with the magic sequence `$1$`, followed by an up to 8 character *salt*. DES hashes start with a 2 character *salt*. Note that a DES hash considers only the first 8 characters. The MD5 considers the whole string.

---

<sup>4</sup>It is unclear to me whether a MIME note can contain a mixture of content and parts, but I believe the answer is 'no'.

Salt and algorithm can be forced by instantiating the start of *Encrypted* with it. This is typically used to force MD5 hashes:

```
?- append("$1$", _, E),
    crypt("My password", E),
    format('~s~n', [E]).
```

```
$1$qdaDeDZn$ZUxSQEESEHIDCHPNc3fxZ1
```

*Encrypted* is always an ASCII string. *Plain* only supports ISO-Latin-1 passwords in the current implementation.

*Plain* is either an atom, SWI-Prolog string, list of characters or list of character-codes. It is not advised to use atoms, as this implies the password will be available from the Prolog heap as a defined atom.

## 9 SHA1 and SHA2 Secure Hash Algorithms

The library `sha` provides *Secure Hash Algorithms* approved by FIPS (*Federal Information Processing Standard*). Quoting Wikipedia: “*The SHA (Secure Hash Algorithm) hash functions refer to five FIPS-approved algorithms for computing a condensed digital representation (known as a message digest) that is, to a high degree of probability, unique for a given input data sequence (the message). These algorithms are called secure because (in the words of the standard), for a given algorithm, it is computationally infeasible 1) to find a message that corresponds to a given message digest, or 2) to find two different messages that produce the same message digest. Any change to a message will, with a very high probability, result in a different message digest.*”

The current library supports all 5 approved algorithms, both computing the hash-key from data and the *hash Message Authentication Code* (HMAC).

Input is text, represented as an atom, packed string object or code-list. Note that these functions operate on byte-sequences and therefore are not meaningful on Unicode text. The result is returned as a list of byte-values. This is the most general format that is comfortably supported by standard Prolog and can easily be transformed in other formats. Commonly used text formats are ASCII created by encoding each byte as two hexadecimal digits and ASCII created using *base64* encoding. Representation as a large integer can be desirable for computational processing.

**sha\_hash(+Data, -Hash, +Options)**

Hash is the SHA hash of Data. *Data* is either an atom, packed string or list of character codes. *Hash* is unified with a list of integers representing the hash. The conversion is controlled by Options:

**algorithm(+Algorithm)**

One of `sha1` (default), `sha224`, `sha256`, `sha384` or `sha512`

**hmac\_sha(+Key, +Data, -HMAC, +Options)**

Quoting Wikipedia: “*A keyed-hash message authentication code, or HMAC, is a type of message authentication code (MAC) calculated using a cryptographic hash function in combination with a secret key. As with any MAC, it may be used to simultaneously verify both the data*



*integrity and the authenticity of a message. Any iterative cryptographic hash function, such as MD5 or SHA-1, may be used in the calculation of an HMAC; the resulting MAC algorithm is termed HMAC-MD5 or HMAC-SHA-1 accordingly. The cryptographic strength of the HMAC depends upon the cryptographic strength of the underlying hash function, on the size and quality of the key and the size of the hash output length in bits.”*

*Key and Data* are either an atom, packed string or list of character codes. *HMAC* is unified with a list of integers representing the authentication code. *Options* is the same as for `sha_hash/3`, but currently only `sha1` and `sha256` are supported.

## 9.1 License terms

The underlying SHA-2 library is an unmodified copy created by Dr Brian Gladman, Worcester, UK. It is distributed under the license conditions below.

The free distribution and use of this software in both source and binary form is allowed (with or without changes) provided that:

1. distributions of this source code include the above copyright notice, this list of conditions and the following disclaimer;
2. distributions in binary form include the above copyright notice, this list of conditions and the following disclaimer in the documentation and/or other associated materials;
3. the copyright holder's name is not used to endorse products built using this software without specific written permission.

ALTERNATIVELY, provided that this notice is retained in full, this product may be distributed under the terms of the GNU General Public License (GPL), in which case the provisions of the GPL apply INSTEAD OF those given above.

## 10 Memory files

The `memfile` provides an alternative to temporary files, intended for temporary buffering of data. Memory files in general are faster than temporary files and do not suffer from security risks or naming conflicts associated with temporary-file management. They do assume proper memory management by the hosting OS and cannot be used to pass data to external processes using a file-name.

There is no limit to the number of memory streams, nor the size of them. However, memory-streams cannot have multiple streams at the same time (i.e. cannot be opened for reading and writing at the same time).

These predicates are first of all intended for building higher-level primitives. See also `sformat/3`, `atom_to_term/3`, `term_to_atom/2` and the XPCE primitive `pce_open/3`.

### **new\_memory\_file**(-Handle)

Create a new memory file and return a unique opaque handle to it.

### **free\_memory\_file**(+Handle)

Discard the memory file and its contents. If the file is open it is first closed.

**open\_memory\_file(+Handle, +Mode, -Stream)**

Open the memory-file. *Mode* is currently one of `read` or `write`. The resulting *Stream* must be closed using `close/1`.

**open\_memory\_file(+Handle, +Mode, -Stream, +Options)**

Open a memory-file as `open_memory_file/3`. Options:

**encoding(+Encoding)**

Set the encoding for a memory file and the created stream. Encoding names are the same as used with `open/4`. By default, memoryfiles represent UTF-8 streams, making them capable of storing arbitrary Unicode text. In practice the only alternative is `octet`, turning the memoryfile into binary mode. Please study SWI-Prolog Unicode and encoding issues before using this option.

**size\_memory\_file(+Handle, -Bytes)**

Return the content-length of the memory-file it *Bytes*. The file should be closed and contain data.

**atom\_to\_memory\_file(+Atom, -Handle)**

Turn an atom into a read-only memory-file containing the (shared) characters of the atom. Opening this memory-file in mode `write` yields a permission error.

**memory\_file\_to\_atom(+Handle, -Atom)**

Return the content of the memory-file in *Atom*.

**memory\_file\_to\_codes(+Handle, -Codes)**

Return the content of the memory-file as a list of character-codes in *Codes*.

## 11 Time and alarm library

The `time` provides timing and alarm functions.

**alarm(+Time, :Callable, -Id, +Options)**

Schedule *Callable* to be called *Time* seconds from now. *Time* is a number (integer or float). *Callable* is called on the next pass through a call- or redo-port of the Prolog engine, or a call to the `PL.handle_signals()` routine from SWI-Prolog. *Id* is unified with a reference to the timer.

The resolution of the alarm depends on the underlying implementation. On Unix systems it is based on `setitimer()`, on Windows on `timeSetEvent()` using a resolution specified at 50 milliseconds.<sup>5</sup> Long-running foreign predicates that do not call `PL.handle_signals()` may further delay the alarm.

*Options* is a list of *Name(Value)* terms. Defined options are:

**remove(Bool)**

If `true` (default `false`), the timer is removed automatically after firing. Otherwise it must be destroyed explicitly using `remove_alarm/1`.

---

<sup>5</sup>BUG: The maximum time for `timeSetEvent()` used by the Windows application is 1000 seconds. Calling with a higher time value raises a `resource_error` exception.

**install(*Bool*)**

If *false* (default *true*), the timer is allocated but not scheduled for execution. It must be started later using `install_alarm/1`.

**alarm(+*Time*, :*Callable*, -*Id*)**

Same as `alarm(Time, Callable, Id, [])`.

**install\_alarm(+*Id*)**

Activate an alarm allocated using `alarm/4` with the option `install(false)`. This is intended to reclaim alarms reliably using `call_cleanup/2`. See the implementation of `call_with_time_limit/2`.

**remove\_alarm(+*Id*)**

Remove an alarm. If it is not yet fired, it will not be fired any more.

**current\_alarm(?*At*, ?:*Callable*, ?*Id*, ?*Status*)**

Enumerate the not-yet-removed alarms. *Status* is one of `done` if the alarm has been called, `next` if it is the next to be fired and `scheduled` otherwise.

**call\_with\_time\_limit(+*Time*, :*Goal*)**

True if *Goal* completes within *Time* seconds. *Goal* is executed as in `once/1`. If *Goal* doesn't complete within *Time* seconds (wall time), exit using the exception `time_limit_exceeded`. See `catch/3`.

Please note that this predicate uses `alarm/4` and therefore is *not* capable to break out of long running goals such as `sleep/1`, blocking I/O or other long-running (foreign) predicates. Blocking I/O can be handled using the `timeout` option of `read_term/3`.

## 12 Limiting process resources

The `rlimit` library provides an interface to the POSIX `getrlimit()/setrlimit()` API that control the maximum resource-usage of a process or group of processes. This call is especially useful for server such as CGI scripts and `inetd`-controlled servers to avoid an uncontrolled script claiming too much resources.

**rlimit(+*Resource*, -*Old*, +*New*)**

Query and/or set the limit for *Resource*. Time-values are in seconds and size-values are counted in bytes. The following values are supported by this library. Please note that not all resources may be available and accessible on all platforms. This predicate can throw a variety of exceptions. In portable code this should be guarded with `catch/3`. The defined resources are:

<code>cpu</code>	CPU time in seconds
<code>fsize</code>	Maximum filesize
<code>data</code>	max data size
<code>stack</code>	max stack size
<code>core</code>	max core file size
<code>rss</code>	max resident set size
<code>nproc</code>	max number of processes
<code>nofile</code>	max number of open files
<code>memlock</code>	max locked-in-memory address

When the process hits a limit POSIX systems normally send the process a signal that terminates it. These signals may be caught using SWI-Prolog's `on_signal/3` primitive. The code below illustrates this behaviour. Please note that asynchronous signal handling is dangerous, especially when using threads. 100% fail-safe operation cannot be guaranteed, but this procedure will inform the user properly 'most of the time'.

```
rlimit_demo :-
    rlimit(cpu, _, 2),
    on_signal(xcpu, _, cpu_exceeded),
    ( repeat, fail ).

cpu_exceeded(_Sig) :-
    format(user_error, 'CPU time exceeded~n', []),
    halt(1).
```

## 13 Installation

### 13.1 Unix systems

Installation on Unix system uses the commonly found *configure*, *make* and *make install* sequence. SWI-Prolog should be installed before building this package. If SWI-Prolog is not installed as `pl`, the environment variable `PL` must be set to the name of the SWI-Prolog executable. Installation is now accomplished using:

```
% ./configure
% make
% make install
```

This installs the foreign libraries in `$PLBASE/lib/$PLARCH` and the Prolog library files in `$PLBASE/library`, where `$PLBASE` refers to the SWI-Prolog 'home-directory'.

## Index

add\_stream\_to\_pool/2, 11  
alarm/3, 19  
alarm/4, 18, 19  
at\_end\_of\_stream/1, 10  
atom\_to\_memory\_file/2, 18  
atom\_to\_term/3, 17  
  
call\_cleanup/2, 19  
call\_with\_time\_limit/2, 19  
catch/3, 19  
cgi *library*, 3  
cgi\_get\_form/1, 13  
cgi\_get\_form/2, 14  
close/1, 18  
close\_stream\_pool/0, 11  
crypt *library*, 1, 3, 15  
crypt/2, 15  
current\_alarm/4, 19  
  
delete\_stream\_from\_pool/1, 11  
detach\_IO/0, 3, 4  
dispatch\_stream\_pool/1, 11  
dup/2, 4  
  
exec/1, 3  
  
files *library*, 7  
fork/1, 3, 4, 8, 10  
fork/2, 3  
format/2, 14  
free\_memory\_file/1, 17  
  
gethostname/1, 10  
  
halt/[0  
    1], 4  
hmac\_sha/4, 16  
http/html\_write *library*, 14  
http/thread\_httpd *library*, 11  
  
install\_alarm/1, 19  
ip/4, 13  
is\_process/1, 7  
  
kill/2, 4  
  
memfile *library*, 3, 17  
  
memory\_file\_to\_atom/2, 18  
memory\_file\_to\_codes/2, 18  
mime *library*, 14  
mime\_parse/2, 14  
  
new\_memory\_file/1, 17  
  
on\_signal/2, 4  
on\_signal/3, 20  
once/1, 19  
open/4, 18  
open\_memory\_file/3, 18  
open\_memory\_file/4, 18  
open\_socket/3, 8  
  
pce\_open/3, 17  
pipe/2, 4  
process\_create/3, 5  
process\_id/1, 6  
process\_id/2, 7  
process\_kill/1, 7  
process\_kill/2, 7  
process\_release/1, 7  
process\_wait/2, 7  
process\_wait/3, 7  
  
read/1, 10  
read\_term/3, 19  
remove\_alarm/1, 18, 19  
rlimit *library*, 19  
rlimit/3, 19  
  
set\_time\_file/3, 7  
sformat/3, 17  
sha *library*, 3, 16  
sha\_hash/3, 16, 17  
size\_memory\_file/2, 18  
sleep/1, 19  
socket *class*, 8  
socket *library*, 3, 8, 11  
stream\_pool\_main\_loop/0, 11  
streampool *library*, 11  
string\_to\_list/2, 12  
  
tcp\_accept/3, 8–10  
tcp\_bind/2, 8, 9

tcp\_close\_socket/1, 8  
tcp\_connect/2, 9  
tcp\_fcntl/3, 9, 11  
tcp\_host\_to\_address/2, 10  
tcp\_listen/2, 9  
tcp\_open\_socket/3, 8, 9  
tcp\_setopt/2, 9  
tcp\_socket/1, 8, 12  
term\_to\_atom/2, 17  
time *library*, 18  
  
udp\_receive/4, 12  
udp\_send/4, 9, 13  
udp\_socket/1, 12  
unix *library*, 1, 3  
  
wait/2, 3, 4  
wait\_for\_input/3, 8, 10–12